Problem 32

We shall say that an n-digit number is pandigital if it makes use of all the digits 1 to n exactly once; for example, the 5-digit number, 15234, is 1 through 5 pandigital.

The product 7254 is unusual, as the identity, 39 × 186 = 7254, containing multiplicand, multiplier, and product is 1 through 9 pandigital.

Find the sum of all products whose multiplicand/multiplier/product identity can be written as a 1 through 9 pandigital.
HINT: Some products can be obtained in more than one way so be sure to only include it once in your sum.

Solution

The multiplied numbers can be at most 4 digits long, since a four-digit number multiplied by a one digit number is at least four digits.

```
In[17]:= fourDigs = Range[1000, 9999];

In[18]:= isAllowed[num_] :=
          Sort@DeleteDuplicates@IntegerDigits[num] === Sort@IntegerDigits[num]

In[19]:= allowFourDigs = Select[fourDigs, isAllowed];

In[20]:= ansFourDigs = Select[Tuples[{allowFourDigs, Range[9]}],
           Sort@Flatten[IntegerDigits /@ {#[[1]] #[[2]], #[[1]], #[[2]]}] ===
             {1, 2, 3, 4, 5, 6, 7, 8, 9} &]

Out[20]= {{1738, 4}, {1963, 4}}

In[21]:= threeDigsOrLess = Range[999];
         allowThreeDigs = Select[threeDigsOrLess, isAllowed];

In[23]:= ansThreeDigs = Select[Tuples[allowThreeDigs, {2}],
           Sort@Flatten[IntegerDigits /@ {#[[1]] #[[2]], #[[1]], #[[2]]}] ===
             {1, 2, 3, 4, 5, 6, 7, 8, 9} &]

Out[23]= {{12, 483}, {18, 297}, {27, 198}, {28, 157}, {39, 186}, {42, 138}, {48, 159},
          {138, 42}, {157, 28}, {159, 48}, {186, 39}, {198, 27}, {297, 18}, {483, 12}}

In[24]:= Total@DeleteDuplicates@Flatten[{Times @@@ ansThreeDigs, Times @@@ ansFourDigs}]

Out[24]= 45 228
```